

ODS GRAPHICS IN SAS 9.1.3

Mirjana Stojanovic, Cancer Center, Duke University Medical Center, Durham, NC

Laura Archer, Cancer Center, Duke University Medical Center, Durham, NC

Donna Hollis Cancer Center, Duke University Medical Center, Durham, NC

ABSTRACT

This paper will present a new feature in SAS® 9.1.3 and detail an experimental procedure in SAS/STAT® which demonstrates the future of graphics in SAS.

Graphics are an essential part of statistical presentation and until now customized graphs were cumbersome to create using SAS/GRAPH®. In SAS 9.1 an experimental extension of the Output Delivery System (ODS) is being used with a number of SAS/STAT procedures to automatically create graphics with the same ease ODS provides in creating tables. In order to create statistical graphs using ODS, include the statement ODS GRAPHICS before your procedure statements and this will invoke the facility. Graphics are then created automatically. ODS statistical graphics do not create output for the LISTING destination. You must send your output to other ODS destination (RTF, HTML, DOCUMENT), ODS Graphics uses Java technology, and you do not need to learn SAS Graph to use this.

The paper will give step-by-step instructions on how to create and use the Output Delivery System (ODS) Graphics by demonstrating the production of statistical graphics, customization of template procedure, changing location of legend and tips on how to produce graphics quickly and efficiently. As an example, we are using the LIFETEST procedure from SAS/STAT. We will detail using ODS graphics in conjunction with this SAS/STAT procedure. Then we will give an overview of graph customization using the Graphical Template Language. Next we will show how we customize the layout and the appearance of the LIFETEST procedure using ODS template. Finally to show the ease and efficiency of the using ODS graphics a comparison of how we produced graphs before will be shown with this new statistical graphics technique.

INTRODUCTION

ODS Statistical Graphics, or ODS graphics for short, is an experimental extension of the Output Delivery System (ODS) first introduced in SAS 9.1.3. ODS graphics can be used with several SAS/STAT procedures, and will allow you to create statistical graphics automatically. Once graphics are created modifications can be made using PROC TEMPLATE. This paper will give a brief overview of ODS graphics, and the use of PROC TEMPLATE to modify graphs.

ODS GRAPHICS

Beginning in SAS 8.0 ODS was introduced to create and manage statistical tables, as well as store and display them in several formats including HTML and RTF. In SAS 9.1.3 ODS was expanded to include a graphical component which will allow you to create and manage statistical graphics in much the same way ODS creates and manages tables. Like ODS, ODS Graphics incorporates procedure options (and defaults) to specify what output will be created. Another similarity between ODS and ODS Graphics is both allow for the specification of an ODS destination; a list of ODS destinations for statistical graphics follows. Another similarity to ODS is that ODS Graphics allows you to use the ODS select and ODS exclude statements to select or exclude graphs from the output. Many other similarities exist and users will find that using ODS Graphics can be done with the same ease and control as ODS.

Table 1. Destinations supported by ODS Graphics:

Destination	Viewer
DOCUMENT	NA
HTML	Browser
LATEX	Ghostview
PCL	Ghostview
PDF	Acrobat
PS	Ghostview
RTF	Microsoft Word

To invoke ODS graphics include the code ODS GRAPHICS ON before your PROC statement. ODS Graphics does not support the LISTING destinations so prior to the ODS GRAPHICS ON statement an ODS destination should be specified, for example:

```
ODS HTML;  
ODS GRAPHICS ON;
```

Here ODS HTML specifies that the output should be stored in the HTML destination, and ODS GRAPHICS ON request ODS graphics be added to the tabular output.

ODS GRAPHICS can be used with the following procedures in SAS 9.1.

Table 2. Procedures Supporting ODS Graphics:


Base SAS	SAS/STAT
CORR	ANOVA
	CORRESP
SAS/ETS	GAM
ARIMA	GENMOD
AUTOREG	GLM
ENTROPY	KDE
EXPAND	LIFETEST
MODEL	LOESS
SYSLIN	LOGISTIC
TIMESERIES	MI
UCM	MIXED
VARMAX	PHREG
X12	PRINCOMP
	PRINQUAL
SAS High-Performance Forecasting	REG
HPF	ROBUSTREG

For the remainder of this paper, we will show all examples using PROC LIFETEST, and the following data. Our data contains 1264 patients with colorectal cancer. We are interested in looking at disease free survival (in years) based on P53 (a biological marker) results.

The following code shows the invocation of the ODS Graphics for our example.

```
ods rtf file = "C:\Pharmasug2007\89803_survQ.rtf";  
ods graphics on;  
  
proc lifetest data=p53seq plots=(s) outsurv=dfsout;  
  time dfstimeyr*dfscens(0);  
run;  
  
ods graphics off;  
ods rtf close;
```

Here the first statement ODS RTF specifies the ODS output type, and the FILE = option specifies the path and file name in which to store the output. Next ODS GRAPHICS ON will add statistical graphics to the output. The next set of code is the simplest code available for the PROC LIFETEST procedure. To disable the graphics after execution, enter the statement ODS GRAPHICS OFF. Then close the listing with the concluding statement ODS RTF CLOSE. Executing the code above will produce the following output:



NOTE: ODS Graphics uses JAVA technology. If you encounter "ERROR: Java class generated an exception" it means version 1.5 of the JRE has been installed on your system. To avoid this error and produce ODS graphics, use the following fixes. If you do not need SUN's Java JRE 1.5.0_02 installed then you uninstall this program using the add/remove programs window from the windows control panel. Then you can install 1.4.1_02 from http://java.sun.com/products/archive/j2se/1.4.1_02. If, however, you need to keep 1.5.0_02 installed, you can get SAS 9.1.3 to work with it by taking the following steps. First, find the file called sasv9.cfg (this is typically found in the folder !sasroot\nls\en, where !sasroot is the root directory in which SAS v9 resides). Before editing sasv9.cfg, copy it and rename one of the files sasv9.old. Second, edit the file that is still called sasv9.cfg. Find the line containing the JREOPTIONS. Edit this line to remove the option: -Xbootclasspath/p:!\sasroot\core\sasmisc\sas.ads.webAFrt.jar. If the -JREOPTION parameter does not contain this parameter skip this step, but continue with the following steps. Still on the JREOPTIONS line, modify the parameter -Dsas.jre=private to read -Dsas.jre=public and then save the file. Finally, find the file !sasroot\core\sasexe\jldplgin.dll and rename it to !sasroot\core\sasexe\jldplgin.old

ODS Graphics can also use procedural options to specify which graphics are included in the output.

```
ods rtf file = "C:\Pharmasug2007\89803_survQ.rtf";
ods graphics on;
ods noptitle;

proc format;
  value p53format
    0 = " Wild Type "
    1 = "All Others";
run;


proc lifetest data=p53seq plots=(s);
  time dfstimeyr*dfscens(0);
  strata p53mut2;
  format p53mut2 p53format.;
  title1 "CALGB 89803 dfs p53 Status ";
```

```
run;

ods rtf close;
ods graphics off;
```

This set of codes begins the same way, by specifying the ODS destination and invoking the ODS procedure. We have added the ODS option ODS NOPTITLE to hide the procedure titles. Next we use the PROC FORMAT section of the code to customize the legend produced the PROC LIFETEST procedure. The PROC LIFETEST procedure specifies survival graphs from the PLOTS = (s) option. The legend and long-rank p-value annotations are automatically generated because of the STRATA specification. We also use the procedural codes to format the graph with the FORMAT statement and add a meaningful title using the TITLE statement. The section of code above produces the following output:

CALGB 89803 dfs p53 Status



In addition to using procedural statements to customize output, you can also use ODS statements. For example, we can use ODS SELECT (or EXCLUDE) to select (or exclude) certain graphs. The following code will show an example of this feature.

```
ods trace on;
ods rtf file = "C:\Pharmasug2007\89803_survQ.rtf";
ods graphics on;
ods noptitle;
ods select survival;

proc lifetest data=p53seq plots=(s);
  time dfstimeyr*dfscens(0);
  strata p53mut2;
  format p53mut2 p53format.;
  title1 "CALGB 89803 dfs p53 Status ";
run;

ods rtf close;
ods graphics off;
ods trace off;
```

This code, while similar to the previous code, has included the ODS SELECT statement. We have also included the ODS TRACE ON statement. This will allow us to see a list of all possible graphics that are produced by the LIFETEST procedure in which we can select to include (or exclude) in the output. Below is a partial list of what is produced in your log from the ODS TRACE ON statement:

```

Output Added:
-----
Name:          WilHomCov
Label:         Wilcoxon Covariance
Template:     Stat.Lifetest.Matrix
Path:         Lifetest.StrataHomogeneity.WilHomCov
-----

```

```

Output Added:
-----
Name:          HomTests
Label:         Homogeneity Tests
Template:     Stat.Lifetest.HomTests
Path:         Lifetest.StrataHomogeneity.HomTests
-----

```

WARNING: Statistical graphics displays created with ODS are experimental in this release.

```

Output Added:
-----
Name:          Survival
Label:         Product-Limit Survival
Template:     Stat.Lifetest.Graphics.ProductLimit
Path:         Lifetest.Survival
-----

```

From this list we see that the graphic we wish to select is called survival. Thus the statement ODS SELECT SURVIVAL specifies that only the survival graphic should be included in the output. The code above will produce the same graph as above, but it will not also produce 22 pages of other output from the LIFETEST procedure.

To summarize:


- ODS Graphics Statement (Experimental) enables automatic graphic capabilities.
- Restrictions to ODS Graphics are:
 - ODS Graphics do not create output for the listing destination. You must send your output to some other ODS Destination (HTML, RTF, PDF) to obtain output.
 - ODS Graphics also do not support any SAS/GRAPH global statements' (Goptions, Symbol, Pattern) as well as GTITLE, and GFOOTNOTE.
- Syntax:
 - **ODS graphics <OFF | ON < / options>>;**
 - Arguments OFF- turns off automatic ODS Graphics generation
 - Arguments ON – turns on the automatic ODS Graphics generation

PROC TEMPLATE


Although ODS Graphics can customize which graphs are included in the output, there may still arise the occasion to customize the appearance of the graphs. We will approach this topic next through the PROC TEMPLATE procedure.

The two major components that govern the appearance of a graphic are the style template and the graph template. The style template controls the presentation appearance for all output objects produced, and provides a consistent appearance between graphical objects. There are four SAS defined style templates available to use with ODS graphics. They are Default, Journal, Analysis, and Statistical. Some of the formatting styles included are font (size, weight and style), line style (color, thickness and style), and marker style. To specify one of these styles, add the style = option in the ODS destination. For example:


```
ods rtf file = "C:\Pharmasug2007\89803_survQ.rtf" style = journal;
```




Style = Default



Style = Journal



Style = Analysis



Style = Statistical

The graph template controls the appearance of individual graphs and can include the formatting of headers and footnotes, addition of title lines, position of the legend, and colors and thickness of graphical lines. Both style templates and graph templates can be customized, however, in this paper we will limit the discussion to the customization of graph templates. The extension to style templates is similar.

Before you can customize graph templates, it is best to understand where templates reside, and how they can be viewed. Templates are stored in a new file type called an item store. Template stores can have read, update, or write access. SAS defines two template stores SASUSER.TEMPLAT (update) and SASHELP.TMPLST (read). Files stored in SASHELP.TMPLST are provided by SAS and are the templates currently being used unless others have been defined. They have read only access and cannot be updated or over written, and new templates cannot be stored here. In contrast SASUSER.TEMPLAT is where user customized templates can be written, updated, and stored.

Since available templates can be stored in more than one template store with the same template name, SAS ODS will search the template stores in a specific order until it finds a template with the appropriate name. To see the order in which template stores will be searched type:

```
ods path show;
```

The following will be written to the log:

```
Current ODS PATH list is:
```

1. SASHELP.TMPLMST(READ)

ODS path can also be used to identify a new template store, define search order, and assign permissions. So, for example, you could assign the following:

```
ODS PATH work.temp (update) sasuser.tempat (update) sashelp.tmplmst (read);
```

This will create a new template store in the library work with update access, and will assign the order in which to search for appropriate templates first in work, then sasuser, and then sashelp. To verify this, re-run the code `ods path show`; The following will now be written to the log:

Current ODS PATH list is:

1. WORK.TEMP(UPDATE)
2. SASUSER.TEMPAT(UPDATE)
3. SASHELP.TMPLMST(READ)


For our purposes, we found it easier to customize an existing template as oppose to creating a new graph template from scratch. Before we can customize a graph template, we need to know which template is being used by default. Recall that using `ODS TRACE ON` prior to the SAS procedure will tell you not only the names of the graphs being use, as detailed earlier in the paper, but it will also tell you the default template. From above, the following was written to the log when we included `ODS TRACE ON` in our code:

```
WARNING: Statistical graphics displays created with ODS are experimental in this release.
```

Output Added:

```
-----  
Name:      Survival  
Label:     Product-Limit Survival  
Template:  Stat.Lifetest.Graphics.ProductLimit  
Path:     Lifetest.Survival  
-----
```

Thus we see that the default template for the survival graph in PROC LIFETEST is `Stat.Lifetest.Graphics.ProductLimit`. Once we have determined the template we want to customize, we must find the source code. Source code can be found in three ways. The first is to right click on the result viewer (on the far left) then scroll down to templates and select the location of the template (if you are customizing a SAS defined default, this will be in the location `SASHELP.TMPLST`). Next, follow the drop down menus to find the procedure you are using. To find `LIFETEST` you would select `Stat`, then `Lifetest`, then `Graphics`. This will show the screen seen below. Finally double click on the template of interest. In our case `ProductLimit`.



The second way to get to the directory of templates is to type *ODSTEMPLATES* in the SAS command window. This will bring up the an explorer window and you will need to follow the steps above to find the template of interest. The third way is to use the source statement in the *TEMPLATE* procedure. For example the following code will show the source code for the template we wish to customize.

```
proc template;
  source stat.lifetest.graphics.productlimit;
run;
```

The source code will print to the SAS log by default, but a file destination can be specified if you wish to store the code in an external file.

Once you have the source code, you must find the section you want to customize. For our example, we wanted to move the location of the legend from the default position (upper right) to any of the four corners. From the log we can copy the code into our editor. A portion of the code follows:

```
define statgraph Stat.Lifetest.Graphics.ProductLimit;
  < MORE SAS SUPPLIED CODE>
  DiscreteLegend "Survival" / hAlign=right vAlign=top across=
    1 border=true;
  < MORE SAS SUPPLIED CODE>
end;
```

To change the legend to the right bottom simple change the code as follows:

```
define statgraph Stat.Lifetest.Graphics.ProductLimit;
  < MORE SAS SUPPLIED CODE>
  DiscreteLegend "Survival" / hAlign=right vAlign=Bottom
    across=1 border=true;
  < MORE SAS SUPPLIED CODE>
end;
```

You must change this code everywhere it occurs. For example in this section of code the definition for the *DiscreteLegend* occurs in three separate places, and the change must be made in each of these three places. Further we wish to store this code in a permanent library so that other users will have access to it.

```
libname NewTemp 'L:\Template_Macro' ;
options MSTORED SASMSTORE = NewTemp;

ODS PATH NewTemp.TEMPLAT SASHELP.TMPLMST;

proc template;
  source stat.lifetest.graphics.productlimit ;
  define statgraph Stat.Lifetest.Graphics.ProductLimit / store = NewTemp.TEMPLAT;
    < MORE SAS SUPPLIED CODE>
  DiscreteLegend "Survival" / hAlign=right vAlign=Bottom
    across=1 border=true;
    < MORE SAS SUPPLIED CODE>
  end;
run;
```

Finally, for ease we have created a macro to specify the location of the legend, and stored this compiled macro in a permanent library using the following code:

```
libname NewTemp 'L:\Template_Macro' ;
options MSTORED SASMSTORE = NewTemp;

%macro LIFETMP (RightLeft=, TopBottom=) / store source;
proc template;
define statgraph Stat.Lifetest.Graphics.ProductLimit / store = NewTemp.TEMPLAT;
  < MORE SAS SUPPLIED CODE>
  DiscreteLegend "Survival" / hAlign=&rightleft vAlign=&topbottom
    across=1 border=true;
  < MORE SAS SUPPLIED CODE>
end;
```

```
run;
mend LIFETEMP;
```

Then to execute this code simply run :

```
libname NewTemp "L:\template_macro\" ;
ods path newtemp.templat sashelp.templmst ;
options mstored sasstore=newtemp;

%lifetmp (RightLeft=right,TopBottom=bottom)

ods graphics on;
ods rtf file = "L:\template_macro\TEST_template.rtf" ;
ods noptitle;
ods select survival ;

proc lifetest data=p53seq(keep=dfstimeyr dfscens p53mut2) plots=(s) outsurv=dfsout;
  time dfstimeyr*dfscens(0);
  strata p53mut2;
  title1 "CALGB 89803 dfs p53 Status " ;
  title2 h=0.75 j=1 "Survival(1) " ;
  title4 "Figure 1: Kaplan-Meier Estimates of Survival Function";
  footnote1 h=0.75 j=1 "Output: E:\Survival\" j=r "[Program:
    89803_dfsallvars_072006.sas]";
run;

ods rtf close;
ods graphics off;
```


Here the LIBNAME statement is assigning to the directory where we stored the macro. Next, we use the ODS PATH statement to assign the paths we wish to search and the order in which to search them. Next, we need to include the options to read the stored macros. Now we are ready to execute the macro. Once we have done this, we can run the remainder of the code as before. This code produces the following output:

03:09 Thursday, February 22, 2007 1

CALGB 89803 dfs p53 Status

Survival(1)

Figure 1: Kaplan-Meier Estimates of Survival Function



For a point of reference, prior to the using and customizing ODS graphic we used SAS to find statistical results and manipulate and store datasets. Then we import the data set into S-Plus® and use S-Plus to create the graph. To create the graphs we either have to know S-Plus, or run other peoples pre-written code or be familiar with interactive S-Plus. So you must have another level of education. Also a separate section of code must be run to produce the legend and the titles. Finally the graphs must be exported into word (or another usable format).

CONCLUSION:

The use of ODS Graphics will allow us to make customized graphics with ease, and not depend on knowing multiple statistical software packages. While the template language is complicated the degree of modification required is often not difficult. By additional use of the macro language can create macros incorporating all of our customizable changes and make them available to all of our department, thus intimate knowledge of PROC TEMPLATE is only required by a few people.

In conclusion ODS Graphics is a wonderful new experimental feature in SAS that will allow the creation of statistical graphics. Graphs will maintain a professional appearance, and with the uses of styles will look consistent with other ODS output. Further, with use of PROC TEMPLATE, graphs have reached a level of user customization that has been difficult to achieve prior to now.

REFERENCES:

1. SAS Institute, Inc. (2002-2004), *SAS/STAT User's Guide, SAS OnlineDoc® 9.1.3*, Cary, NC: SAS Institute, Inc.
2. SAS Institute, Inc. (2005), "TEMPLATE Procedure: Creating ODS Statistical Graphics Output (Experimental)"
3. SAS Institute, Inc. (2005), "ODS Statistical Graphics (Experimental) and ODS Styles: Usage and Reference"
4. Cheng, Wei (2005), "SAS 8 Is Seen in the Rearview Mirror: Generate Statistical Graphics Using ODS in SAS® 9" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
5. Cartier, Jeff (2005), "The Power of the Graphics Template Language" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
6. Rodriguez, Robert N. (2004), "An Introduction to ODS for Statistical Graphics in SAS® 9.1" *Proceedings of the 29th Annual SAS Users Group International (SUGI) Conference*.
7. Zender, Cynthia (2005) "Customizing ODS Statistical Graphs" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
8. Jeff Cartier, (2006) "A Programmer's Introduction to the Graphics Template Language" *Proceedings of the 31th Annual SAS Users Group International (SUGI) Conference*.
9. Mitchell, Rick (2004) "Thumbs Up For ODS Graphics, But Don't Throw Out All Your SAS/GRAPH Programs!" *Proceedings of the 29th Annual SAS Users Group International (SUGI) Conference*.
10. Rodriguez, Robert N.(2006) "Creating Statistical Graphics in SAS 9.2: What Every Statistical User Should Know" *Proceedings of the 31th Annual SAS Users Group International (SUGI) Conference*
11. Shi-Tao Yeh (2005) "Customizing ODS Statistical Graphics" Graphs "Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference.
12. Shi-Tao Yeh (2005) "ODS and Statistical Graphics" Graphs "NESIG17"
13. Cheng, Wei (2006), "ODS Statistical Graphics for Clinical Rresearch" *Proceedings of the 31th Annual SAS Users Group International (SUGI) Conference*.
14. Cartier, Jeff (2005) "Use of Styles in Graphics" *Proceedings of the 30th Annual SAS Users Group International (SUGI) Conference*.
15. McNeill, Sandy (2002) "What's New in the Output Delivery System, Version 9.0" *The fifteen annual NESUG Conference*
16. SAS 9.1.3 Output Delivery System: User's Guide
http://support.sas.com/documentation/onlinedoc/91pdf/sasdoc_913/base_ods_9268.pdf

<http://support.sas.com/rnd/base/topics/statgraph/index.html>

http://support.sas.com/91doc/getDoc/statug.hlp/odsgraph_sect16.htm

CONTACT INFORMATION:

Your comments are greatly appreciated and encouraged. Contact the author at:

Mirjana Stojanovic
Duke University Medical Center
(919)668-9337(phone)
e-mail: mirjana.stojanovic@duke.edu

Donna Hollis
Duke University Medical Center
(919) 681-5027(phone)
e-mail: donna.hollis@duke.edu

Laura Archer
Duke University Medical Center
e-mail laura.archer@duke.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.